Pensieve Header: Wheeled Semi-Symmetrized calculus in the 2D quotient.

Continues “Semi-Symmetrized 2D Calculus.nb”, “A 2D B-Picture, IV.nb”, uses computations from “BCH in Blobs.nb” and from “111223 Calculator.nb”, “111227 Calculator.nb”, “111228 Calculator.nb”, and “111230 Calculator.nb”. Continued in “2012-01/betaCalculus.nb”.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\w-Computations"]
```

```
C:\drorbn\AcademicPensieve\Projects\w-Computations
```

```
ar[i_, j_] := t[i] h[j]; R[i_, j_] := W[1] + ar[i, j] (E^c[i] - 1) / c[i];
```

```
μCollect[μ_] :=
  Collect[μ, _h, Collect[#, _t, Simplify] &] /. W[ws_] :> W[Simplify[ws]];
SetAttributes[μForm, Listable];
μForm[μ_] := Module[
  {tails, heads, mat},
  tails = Union[Cases[μ, t[s_] :> s, Infinity]];
  heads = Union[Cases[μ, h[s_] :> s, Infinity]];
  mat = Outer[FullSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
  PrependTo[mat, t /@ tails];
  mat = Prepend[Transpose[mat],
    Prepend[h /@ heads, μ /. (h[_] | t[_]) → 0]
   ];
  MatrixForm[mat]
 ]
```

Tails Works

```
tm[x_, y_, z_][μ_] := μCollect[μ /. {t[x] | t[y] → t[z], c[x] | c[y] → c[z]}];
tΔ[z_, x_, y_][μ_] := μCollect[μ /. {t[z] → t[x] + t[y], c[z] → c[x] + c[y]}];
tS[x_][μ_] := μCollect[μ /. {t[x] → -t[x], c[x] → -c[x]}];
```

Heads Works

```
hm[x_, y_, z_][μ_] := Module[
   {ξ, η},
   ξ = D[μ, h[x]];
   η = D[μ, h[y]];
   μCollect[(μ /. h[x | y] → 0) + ξ h[z] + (1 + ξ /. t[s_] :> c[s]) η h[z]]
  ];
hΔ[z_, x_, y_][μ_] := μCollect[μ /. h[z] → h[x] + h[y]];
hS[x_][μ_] := Module[{β},
   β = 1 + D[μ, h[x]] /. t[s_] :> c[s];
   μCollect[μ /. h[x] → -h[x] / β]
  ];
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]]
```

```
h[5] (t[1] + (1 + c[1]) t[2])
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

0	h[5]
t[1]	1
t[2]	1 + c[1]

Associativity of Heads Multiplication

```
μ1 = α1 ar[1, 1] + α2 ar[2, 2] + α3 ar[3, 3]
```

α1 h[1] t[1] + α2 h[2] t[2] + α3 h[3] t[3]

```
μ1 // μForm
```

0	h[1]	h[2]	h[3]
t[1]	α1	0	0
t[2]	0	α2	0
t[3]	0	0	α3

```
μ1 // hm[1, 2, 1] // μForm
```

0	h[1]	h[3]
t[1]	α1	0
t[2]	α2 + α1 α2 c[1]	0
t[3]	0	α3

```
(t1 = μ1 // hm[1, 2, 1] // hm[1, 3, 1]) // μForm
```

0	h[1]
t[1]	α1
t[2]	α2 + α1 α2 c[1]
t[3]	α3 (1 + α1 c[1]) (1 + α2 c[2])

```
(t2 = μ1 // hm[2, 3, 2] // hm[1, 2, 1]) // μForm
```

0	h[1]
t[1]	α1
t[2]	α2 + α1 α2 c[1]
t[3]	(1 + α1 c[1]) (α3 + α2 α3 c[2])

```
(t1 - t2 ) // μForm
```

0	h[1]
t[1]	0
t[2]	0
t[3]	0

Compatibility of m and Δ

```
μ = α1 ar[1, 1] + α2 ar[2, 2];
{
  μ // hΔ[1, 1, 3] // hΔ[2, 2, 4] // hm[1, 2, 1] // hm[3, 4, 2],
  μ // hm[1, 2, 1] // hΔ[1, 1, 2]
 } // μForm
```

{

0	h[1]	h[2]
t[1]	α1	α1
t[2]	α2 + α1 α2 c[1]	α2 + α1 α2 c[1]

,

0	h[1]	h[2]
t[1]	α1	α1
t[2]	α2 + α1 α2 c[1]	α2 + α1 α2 c[1]

}

▪ The Antipode Property

```
{
 α ar[1, 1] // hΔ[1, 1, 2] // hS[2] // hm[1, 2, 1],
 α ar[1, 1] // hΔ[1, 1, 2] // hS[2] // hm[2, 1, 1]
}
```

```
{0, 0}
```

Factorization

```
hfac[z_, xtails_List → x_, y_][μ_] := Module[
    {ytails},
    ytails = Complement[
      Union[Cases[μ, t[s_] ⧴ s, Infinity]],
      xtails
     ];
    hfac[z, xtails → x, ytails → y][μ]
   ];
hfac[z_, x_,  ytails_List → y_][μ_] := Module[
    {xtails},
    xtails = Complement[
      Union[Cases[μ, t[s_] ⧴ s, Infinity]],
      ytails
     ];
    hfac[z, xtails → x, ytails → y][μ]
   ];
hfac[z_, xtails_List → x_, ytails_List → y_][μ_] := Module[
   {ζ, ξ, η},
   ζ = D[μ, h[z]];
   ξ = ζ /. ((t[#] → 0) & /@ ytails);
   η = ζ /. ((t[#] → 0) & /@ xtails);
   μCollect[μ - h[z] ζ + h[x] ξ + h[y] η / (1 + ξ /. t[s_] → c[s])]
  ]
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

$$\begin{pmatrix} 0 & h[5] \\ t[1] & 1 \\ t[2] & 1 + c[1] \end{pmatrix}$$

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // hfac[5, {1} → 3, 4]
```

```
h[3] t[1] + h[4] t[2]
```

Conjugation

```
conj[y_, x_][μ_] := Module[
   {ν, x0, x1, γ, ξ, η, a},
   ν = μ // hfac[x, {y} → x0, x1];
   γ = Coefficient[ν, ar[y, x0]];
   ν = μCollect[ν /. W[ws_] :→ W[ws * (c[y] γ + 1)]];
   ξ = D[ν, h[x1]];
   η = D[ν, t[y]];
   a = 1 + ξ /. t[s_] :→ c[s];
   ν = μCollect[(ν /. t[y] → a t[y]) - c[y] ξ η];
   ν // hm[x0, x1, x]
  ];
conji[y_, x_][μ_] := μ // hS[x] // conj[y, x] // hS[x];
```

```
{
  μ1 = W[1] + α ar[1, 1] + β ar[1, 2] + γ ar[2, 1] + δ ar[2, 2],
  μ1 // conj[1, 2],
  μ1 // conji[1, 2],
  μ1 // conj[1, 2] // conji[1, 2]
 } // μForm
```

$$\left\{\begin{pmatrix} W[1] & h[1] & h[2] \\ t[1] & \alpha & \beta \\ t[2] & \gamma & \delta \end{pmatrix}, \begin{pmatrix} W[1+\beta\, c[1]] & h[1] & h[2] \\ t[1] & \alpha+\frac{\alpha\,\delta\, c[2]}{1+\beta\, c[1]} & \beta+\frac{\beta\,\delta\, c[2]}{1+\beta\, c[1]} \\ t[2] & \gamma-\frac{\alpha\,\delta\, c[1]}{1+\beta\, c[1]} & \frac{\delta}{1+\beta\, c[1]} \end{pmatrix},\right.$$

$$\left.\begin{pmatrix} W\left[\frac{1+\delta\, c[2]}{1+\beta\, c[1]+\delta\, c[2]}\right] & h[1] & h[2] \\ t[1] & \frac{\alpha}{1+\delta\, c[2]} & \frac{\beta}{1+\delta\, c[2]} \\ t[2] & \gamma+\frac{\alpha\,\delta\, c[1]}{1+\delta\, c[2]} & \delta+\frac{\beta\,\delta\, c[1]}{1+\delta\, c[2]} \end{pmatrix}, \begin{pmatrix} W[1] & h[1] & h[2] \\ t[1] & \alpha & \beta \\ t[2] & \gamma & \delta \end{pmatrix}\right\}$$

```
(μ2 = W[1] + α1 ar[1, 1] + α2 ar[2, 1] + α3 ar[2, 2] + α4 ar[2, 3]) // μForm;
(μ2 = W[1] + Sum[αi,j ar[i, j], {i, 2}, {j, 3}]) // μForm
```

$$\begin{pmatrix} W[1] & h[1] & h[2] & h[3] \\ t[1] & \alpha_{1,1} & \alpha_{1,2} & \alpha_{1,3} \\ t[2] & \alpha_{2,1} & \alpha_{2,2} & \alpha_{2,3} \end{pmatrix}$$

```
μ2 // conj[1, 2] // μForm
```

$$\begin{pmatrix} W[1+c[1]\,\alpha_{1,2}] & h[1] & h[2] & h[3] \\ t[1] & \alpha_{1,1}\left(1+\frac{c[2]\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}}\right) & \alpha_{1,2}\left(1+\frac{c[2]\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}}\right) & \alpha_{1,3}\left(1+\frac{c[2]\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}}\right) \\ t[2] & \alpha_{2,1}-\frac{c[1]\,\alpha_{1,1}\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}} & \frac{\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}} & -\frac{c[1]\,\alpha_{1,3}\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}}+\alpha_{2,3} \end{pmatrix}$$

```
(t1 = μ2 // conj[1, 2] // conj[1, 3] // hm[2, 3, 2]) // μForm
```

$$\begin{pmatrix} W\left[(1+c[1]\,\alpha_{1,2})\left(1+c[1]\,\alpha_{1,3}\left(1+\frac{c[2]\,\alpha_{2,2}}{1+c[1]\,\alpha_{1,2}}\right)\right)\right] & h[1] \\ t[1] & \frac{\alpha_{1,1}\,(1+c[1]\,\alpha_{1,2}+c[2]\,\alpha_{2,2})\,(1+c[1]\,\alpha_{1,3}+c[\cdots}{(1+c[1]\,\alpha_{1,2})\,(1+c[1]\,\alpha_{1,3})+c[1]\,c[2]\,\alpha_{1,\cdots}} \\ t[2] & \frac{\alpha_{2,1}\,((1+c[1]\,\alpha_{1,2})\,(1+c[1]\,\alpha_{1,3})+c[1]\,c[2]\,\alpha_{1,3}\,\alpha_{2,2})-c[1]\,\alpha_{1,1}\,(\alpha_{2,\cdots}}{(1+c[1]\,\alpha_{1,2})\,(1+c[1]\,\alpha_{1,3})+c[1]\,c[2]\,\alpha_{1,\cdots}} \end{pmatrix}$$

```
(t2 = μ2 // hm[2, 3, 2] // conj[1, 2]) // μForm
```

$$\begin{pmatrix} W[1+c[1](\alpha_{1,2}+\alpha_{1,3}(1+c[1]\alpha_{1,2}+c[2]\alpha_{2,2}))] & h[1] & \\ t[1] & \alpha_{1,1}\left(1+\frac{c[2](\alpha_{2,2}+(1+c[1]\alpha_{1,2}+c[2]\alpha_{2,2})\alpha_{2,3})}{1+c[1](\alpha_{1,2}+\alpha_{1,3}(1+c[1]\alpha_{1,2}+c[2]\alpha_{2,2}))}\right) & (\alpha_{1,2} \\ t[2] & \alpha_{2,1}-\frac{c[1]\alpha_{1,1}(\alpha_{2,2}+(1+c[1]\alpha_{1,2}+c[2]\alpha_{2,2})\alpha_{2,3})}{1+c[1](\alpha_{1,2}+\alpha_{1,3}(1+c[1]\alpha_{1,2}+c[2]\alpha_{2,2}))} & \end{pmatrix}$$

```
Simplify[t1 == t2]
```

True

```
Simplify[
 (μ2 // conji[1, 2] // conji[1, 3] // hm[3, 2, 2]) == (μ2 // hm[3, 2, 2] // conji[1, 2])]
```

True

■ "4T"

```
Riffle[
 ComposeList[
   ops = {conj[2, 1], hΔ[1, 1, 3], hm[2, 3, 2], hΔ[1, 1, 3], hS[3], hm[3, 2, 2]},
   α1 ar[1, 1] + α2 ar[2, 2]
  ] // μForm,
 ops
]
```

$$\left\{\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \alpha_2 \end{pmatrix}, \text{conj}[2,1], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & -c[2]\alpha_1\alpha_2 \\ t[2] & 0 & (1+c[1]\alpha_1)\alpha_2 \end{pmatrix},\right.$$

$$\text{h}\triangle[1,1,3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & -c[2]\alpha_1\alpha_2 & \alpha_1 \\ t[2] & 0 & (1+c[1]\alpha_1)\alpha_2 & 0 \end{pmatrix}, \text{hm}[2,3,2],$$

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_1 \\ t[2] & 0 & (1+c[1]\alpha_1)\alpha_2 \end{pmatrix}, \text{h}\triangle[1,1,3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_1 & \alpha_1 \\ t[2] & 0 & (1+c[1]\alpha_1)\alpha_2 & 0 \end{pmatrix},$$

$$\left.\text{hS}[3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_1 & -\frac{\alpha_1}{1+c[1]\alpha_1} \\ t[2] & 0 & (1+c[1]\alpha_1)\alpha_2 & 0 \end{pmatrix}, \text{hm}[3,2,2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \alpha_2 \end{pmatrix}\right\}$$

```
α1 ar[1, 1] + α2 ar[2, 2] (Exp[c[2]] - 1) / c[2] // conj[2, 1] // hΔ[1, 1, 3] //
    hm[2, 3, 2] // hΔ[1, 1, 3] // hS[3] // hm[3, 2, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \frac{(-1+e^{c[2]})\alpha_2}{c[2]} \end{pmatrix}$$

```
Riffle[
 ComposeList[
   ops = {hΔ[1, 1, 3], hm[2, 3, 2], hΔ[1, 1, 3], hS[3], hm[3, 2, 2]},
   α1 ar[1, 1] + α2 ar[1, 2]
  ] // μForm,
 ops
]
```

$$\left\{\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2 \end{pmatrix}, h\triangle[1,1,3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2 & \alpha_1 \end{pmatrix}, hm[2,3,2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2+\alpha_1\,(1+c[1]\,\alpha_2) \end{pmatrix}, h\triangle[1,1,3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2+\alpha_1\,(1+c[1]\,\alpha_2) & \alpha_1 \end{pmatrix}, hS[3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2+\alpha_1\,(1+c[1]\,\alpha_2) & -\frac{\alpha_1}{1+c[1]\,\alpha_1} \end{pmatrix}, hm[3,2,2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2 \end{pmatrix}\right\}$$

The Double

```
dm[x_, y_, z_][μ_] := μ // conji[y, x] // hm[x, y, z] // tm[x, y, z];
dΔ[z_, x_, y_][μ_] := μ // hΔ[z, x, y] // tΔ[z, x, y];
dS[x_][μ_] := μ // tS[x] // conj[x, x] // hS[x];
```

```
Unprotect[NonCommutativeMultiply];
μ_ ** ν_ := Module[
  {ρ, σ, labels},
  ρ = μ + (ν /. {h[s_] :→ h[σ[s]], t[s_] :→ t[σ[s]], c[s_] :→ c[σ[s]]});
  ρ = ρ /. {
     2 W[a_] :→ W[Simplify[a^2]],
     W[a_] + W[b_] :→ W[Simplify[a * b]]
    };
  labels = Union[Cases[{μ, ν}, h[s_] | t[s_] | c[s_] :→ s, Infinity]];
  Do[
   ρ = ρ // dm[s, σ[s], s],
   {s, labels}
  ];
  ρ
 ]
```

```
ar[1, 2] ** ar[1, 3] // μForm
```

$$\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1 \end{pmatrix}$$

```
ar[1, 3] ** ar[2, 3] // μForm
```

$$\begin{pmatrix} 0 & h[3] \\ t[1] & 1 \\ t[2] & 1+c[1] \end{pmatrix}$$

```
ar[1, 2] ** ar[2, 3] // μForm
```

$$\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & \frac{c[2]}{1+c[1]} \\ t[2] & 0 & \frac{1}{1+c[1]} \end{pmatrix}$$

```
{ar[1, 2] ** ar[1, 3] ** ar[2, 3], ar[2, 3] ** ar[1, 3] ** ar[1, 2]} // μForm
```

$$\left\{\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1+c[2] \\ t[2] & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1+c[2] \\ t[2] & 0 & 1 \end{pmatrix}\right\}$$

```
{R[1, 2] ** R[1, 3] ** R[2, 3], R[2, 3] ** R[1, 3] ** R[1, 2]} // μForm
```

$$\left\{\begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{e^{c[2]}\left(-1+e^{c[1]}\right)}{c[1]} \\ t[2] & 0 & \frac{-1+e^{c[2]}}{c[2]} \end{pmatrix}, \begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{e^{c[2]}\left(-1+e^{c[1]}\right)}{c[1]} \\ t[2] & 0 & \frac{-1+e^{c[2]}}{c[2]} \end{pmatrix}\right\}$$

```
ar[1, 3] ** ar[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1 \end{pmatrix}$$

```
ar[2, 3] ** ar[1, 3] // μForm
```

$$\begin{pmatrix} 0 & h[3] \\ t[1] & 1+c[2] \\ t[2] & 1 \end{pmatrix}$$

```
ar[2, 3] ** ar[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 0 \\ t[2] & 0 & 1 \end{pmatrix}$$

```
{ar[1, 2] // dS[1] , ar[1, 2] // dS[2]}
```

$$\left\{-h[2]\,t[1],\ -\frac{h[2]\,t[1]}{1+c[1]}\right\}$$

```
{R[1, 2], R[1, 2] // dS[1] , R[1, 2] // dS[2]}
```

$$\left\{\frac{\left(-1+e^{c[1]}\right)h[2]\,t[1]}{c[1]}+W[1],\ \frac{\left(-1+e^{-c[1]}\right)h[2]\,t[1]}{c[1]}+W[1],\ -\frac{\left(1-e^{-c[1]}\right)h[2]\,t[1]}{c[1]}+W[1]\right\}$$

```
(R[1, 2] // dS[2]) ** R[1, 2]
```

$W[1]$

```
(R[1, 2] // dS[1]) ** R[1, 2]
```

$W[1]$

```
{R[2, 3] ** dΔ[2, 2, 3][R[1, 2]], dΔ[2, 2, 3][R[1, 2]] ** R[2, 3]} // μForm
```

$$\left\{\begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{e^{c[2]}\left(-1+e^{c[1]}\right)}{c[1]} \\ t[2] & 0 & \frac{-1+e^{c[2]}}{c[2]} \end{pmatrix}, \begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{e^{c[2]}\left(-1+e^{c[1]}\right)}{c[1]} \\ t[2] & 0 & \frac{-1+e^{c[2]}}{c[2]} \end{pmatrix}\right\}$$

```
{ar[2, 3] ** dΔ[2, 2, 3][ar[1, 2]], dΔ[2, 2, 3][ar[1, 2]] ** ar[2, 3]} // μForm
```

$$\left\{\begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1+c[2] \\ t[2] & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & h[2] & h[3] \\ t[1] & 1 & 1+c[2] \\ t[2] & 0 & 1 \end{pmatrix}\right\}$$

```
{R[1, 2] ** dΔ[1, 1, 2][R[1, 3]], dΔ[1, 1, 2][R[1, 3]] ** R[1, 2]} // μForm
```

$$\left\{\begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{e^{-c[1]}\left(-1+e^{c[1]+c[2]}\right)\left(-c[2]+e^{c[1]}\,(c[1]+c[2])\right)}{c[1]\,(c[1]+c[2])} \\ t[2] & 0 & \frac{-e^{-c[1]}+e^{c[2]}}{c[1]+c[2]} \end{pmatrix}, \begin{pmatrix} W[1] & h[2] & h[3] \\ t[1] & \frac{-1+e^{c[1]}}{c[1]} & \frac{-1+e^{c[1]+c[2]}}{c[1]+c[2]} \\ t[2] & 0 & \frac{-1+e^{c[1]+c[2]}}{c[1]+c[2]} \end{pmatrix}\right\}$$

```
(V = W[ω] + α ar[1, 1] + β ar[1, 2] + γ ar[2, 1] + δ ar[2, 2]) // μForm
```

$$\begin{pmatrix} W[\omega] & h[1] & h[2] \\ t[1] & \alpha & \beta \\ t[2] & \gamma & \delta \end{pmatrix}$$

```
{dΔ[1, 1, 2][ar[1, 3]] ** V, V ** ar[1, 3] ** ar[2, 3]} // μForm
```

$$\left\{\begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & 1 \\ t[2] & \gamma & \delta & 1 \end{pmatrix}, \begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & \frac{1+\alpha\, c[1]}{1+\alpha\, c[1]+\gamma\, c[2]}+\frac{\beta\,(1+c[1])\, c[2]}{1+\beta\, c[1]+\delta\, c[2]} \\ t[2] & \gamma & \delta & \frac{\gamma\, c[1]}{1+\alpha\, c[1]+\gamma\, c[2]}+\frac{(1+c[1])\,(1+\delta\, c[2])}{1+\beta\, c[1]+\delta\, c[2]} \end{pmatrix}\right\}$$

```
SolveAlways[
 dΔ[1, 1, 2][ar[1, 3]] ** V == V ** ar[1, 3] ** ar[2, 3],
 {h[1], h[2], h[3], t[1], t[2], c[1], c[2]}
]
```

{}

```
{dΔ[1, 1, 2][R[1, 3]] ** V, V ** R[1, 3] ** R[2, 3]} // μForm
```

$$\left\{\begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & \frac{-1+e^{c[1]+c[2]}}{c[1]+c[2]} \\ t[2] & \gamma & \delta & \frac{-1+e^{c[1]+c[2]}}{c[1]+c[2]} \end{pmatrix}, \begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & \frac{\left(-1+e^{c[1]}\right)(1+\alpha\, c[1])}{c[1]\,(1+\alpha\, c[1]+\gamma\, c[2])}+\frac{e^{c[1]}\left(-1+e^{c[2]}\right)\beta}{1+\beta\, c[1]+\delta\, c[2]} \\ t[2] & \gamma & \delta & \frac{\left(-1+e^{c[1]}\right)\gamma}{1+\alpha\, c[1]+\gamma\, c[2]}+\frac{e^{c[1]}\left(-1+e^{c[2]}\right)(1+\delta\, c[2])}{c[2]\,(1+\beta\, c[1]+\delta\, c[2])} \end{pmatrix}\right\}$$

```
{dΔ[1, 1, 2][R[1, 3]] ** V, V ** R[1, 3] ** R[2, 3]} /.
  {E^c[1] → X, E^c[2] → Y, E^(c[1] + c[2]) → X * Y} // μForm
```

$$\left\{\begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & \frac{-1+X\,Y}{c[1]+c[2]} \\ t[2] & \gamma & \delta & \frac{-1+X\,Y}{c[1]+c[2]} \end{pmatrix}, \begin{pmatrix} W[\omega] & h[1] & h[2] & h[3] \\ t[1] & \alpha & \beta & \frac{(-1+X)\,(1+\alpha\, c[1])}{c[1]\,(1+\alpha\, c[1]+\gamma\, c[2])}+\frac{X\,(-1+Y)\,\beta}{1+\beta\, c[1]+\delta\, c[2]} \\ t[2] & \gamma & \delta & \frac{(-1+X)\,\gamma}{1+\alpha\, c[1]+\gamma\, c[2]}+\frac{X\,(-1+Y)\,(1+\delta\, c[2])}{c[2]\,(1+\beta\, c[1]+\delta\, c[2])} \end{pmatrix}\right\}$$

```
sols = SolveAlways[
  dΔ[1, 1, 2][R[1, 3]] ** V == V ** R[1, 3] ** R[2, 3] /.
   {E^c[1] → X, E^c[2] → Y, E^(c[1] + c[2]) → X * Y},
  {h[1], h[2], h[3], t[1], t[2], c[1], c[2], X, Y}
 ]
```

{}